
Abstract 
 

A new Text File Reader is presented which either outputs a given text or ASCII file line-
by-line to the screen under user control, or dumps the entire contents of the file to the printer in 
one continuous outflow.  

 
——  
 
This paper and its associated code are available online at:  
 
  http://www.bds-soft.com/cocoPapers.php . 
 
===== 

Introduction 
 

What ?!?  
 
Why another text file reader when so many different word processors and similar 

programs are already available? For example: 
 

Telewriter-64 and -128, C. C. Writer (Oh… Um… Yeah, it only 
took me about 35 years to see the “See See Rider” song reference 
(sigh!)), Super Color Writer, V.I.P. Writer, Window Writer, etc. 

 
The short answer is simple: Because I had fun programming it! 
 
The slightly more complex answer is that if I can read a text file, then I can go on to build 

methods that involve checking the text, comparing files, sorting the text, altering the text, or 
taking other actions based on what the text says, i.e. I can build my own text processor, 
assembler, or whatever. More fun! (How about a cross-compiler that runs on the CoCo and 
produces executables that run on the Raspberry Pi? Gack! What have I done?)  

 
——— 

 
A Note on Numbers: To keep everything simple to understand, and also neatly lined-up, I 

frequently refer to numbers as decimal bytes with three full digits, e.g. 004, 027, 229, etc.  See 
Appendix A for conversions between the decimal and hexadecimal representations of bytes. 

 
In works of this complexity (at least for me) typos and other errors are bound to sneak in. 

Please let me know about any you discover so I can note and correct them. 
 
M.D.J. 2021/09/29 
info@bds-soft.com 
 
===== 

TXTFILRD.BAS:  

The Text File Reader  
 

The BASIC Language Program: 
 

1000 '***** 
1010 '* 
1020 '* TXTFILRD.BAS 
1030 '* MDJ 2021/10/16 
1040 '* 
1050 '* TEXT FILE 
1060 '* READER 
1070 '* 
1080 '***** 
1090 ' 
1200 '***** 
1210 '* 
1220 '* INITIAL SETUP 
1230 '* 
1240 '***** 
1250 ' 
1260 PCLEAR 1 
1270 CLEAR 8192 
1280 ' 
1400 '***** 
1410 '* 
1420 '* SPLASH SCREEN 
1430 '* 
1440 '* A$ = USER RESPONSE 
1450 '* 
1460 '***** 
1470 ' 
1480 CLS 
1490 PRINT@ 138, "TXTFILRD.BAS" 
1500 PRINT@ 168, "TEXT FILE READER" 
1510 PRINT@ 262, "BY  M. DAVID JOHNSON" 
1520 PRINT@ 299, "2021/10/16" 
1530 PRINT@ 386, "(PRESS ANY KEY  TO CONTINUE)" 
1540 A$=INKEY$ 
1550 IF A$="" GOTO 1540 
1560 ' 
1700 '***** 
1710 '* 
1720 '* SELECT DISK DRIVE 
1730 '* 

1740 '* DN = DISK DRIVE # 
1750 '* 
1760 '***** 
1770 ' 
1780 CLS 
1790 PRINT@ 66, "PLEASE CHOOSE:" 
1800 PRINT@ 130, "(0) DISK DRIVE 0" 
1810 PRINT@ 162, "(1) DISK DRIVE 1" 
1820 PRINT@ 194, "(2) DISK DRIVE 2" 
1830 PRINT@ 226, "(3) DISK DRIVE 3" 
1840 A$=INKEY$ 
1850 IF A$="" GOTO 1840 
1860 IF A$="0" GOTO 1920 
1870 IF A$="1" GOTO 1940 
1880 IF A$="2" GOTO 1960 
1890 IF A$="3" GOTO 1980 
1900 PRINT@ 386, "(PLEASE PRESS 0, 1, 2, OR 3)" 
1910 GOTO 1840 
1920 DN=0 
1930 GOTO 1990 
1940 DN=1 
1950 GOTO 1990 
1960 DN=2 
1970 GOTO 1990 
1980 DN=3 
1990 PRINT@ 290, "DISK DRIVE#";DN 
2000 PRINT@ 322, "HAS BEEN SELECTED" 
2010 'SET THE DRIVE 
2020 DRIVE DN 
2030 PRINT@ 386, "(PRESS ANY KEY TO CONTINUE)" 
2040 A$=INKEY$ 
2050 IF A$="" GOTO 2040 
2060 ' 
2070 CLS 
2080 PRINT@ 66, "PROCESSING" 
2090 PRINT@ 130, "(PLEASE WAIT)" 
2100 PRINT@ 194, "*"; 
2110 ' 
2200 '***** 
2210 '* 
2220 '* READ THE 
2230 '* DISK DIRECTORY 
2240 '* 
2250 '* TN = TRACK NUMBER 
2260 '* SN = SECTOR NUMBER 
2270 '* B$ = FIRST DISK BUFFER 
2280 '* C$ = SECOND DISK BUFFER 



7 
 

2290 '* 
2300 '***** 
2310 ' 
2320 DIM B$(12), C$(12) 
2330 TN=17 
2340 ' 
2350 'STEP THROUGH THE 
2360 'PERTINENT SECTORS 
2370 FOR SN=3 TO 11 
2380 DSKI$ DN, TN, SN, B$(SN), C$(SN) 
2390 ' 
2400 'PROGRESS COUNTING 
2410 PRINT "*"; 
2420 NEXT SN 
2430 PRINT "*"; 
2440 ' 
2600 '***** 
2610 '* 
2620 '* BREAK THE DISK 
2630 '* BUFFERS INTO 
2640 '* RAW DIRECTORY 
2650 '* BUFFERS 
2660 '* 
2670 '* R$ = RAW BUFFER 
2680 '* T1 = TEMP VARIABLE 
2690 '* 
2700 '***** 
2710 ' 
2720 DIM R$(72) 
2730 ' 
2740 'STEP THROUGH THE 
2750 'PERTINENT SECTORS 
2760 T1 = -1 
2770 FOR SN = 3 TO 11 
2780 T1=T1+1 
2790 R$(T1)=MID$(B$(SN), 1, 32) 
2800 T1=T1+1 
2810 R$(T1)=MID$(B$(SN), 33, 32) 
2820 T1=T1+1 
2830 R$(T1)=MID$(B$(SN), 65, 32) 
2840 T1=T1+1 
2850 R$(T1)=MID$(B$(SN), 97, 32) 
2860 T1=T1+1 
2870 R$(T1)=MID$(C$(SN), 1, 32) 
2880 T1=T1+1 
2890 R$(T1)=MID$(C$(SN), 33, 32) 
2900 T1=T1+1 

2910 R$(T1)=MID$(C$(SN), 65, 32) 
2920 T1=T1+1 
2930 R$(T1)=MID$(C$(SN), 97, 32) 
2940 ' 
2950 'PROGRESS COUNTING 
2960 PRINT "*"; 
2970 NEXT SN 
2980 PRINT@ 226, "*"; 
2990 ' 
3100 '***** 
3110 '* 
3120 '* GET DIRECTORY 
3130 '* LISTING ENTRIES 
3140 '* 
3150 '* F$ = FILENAME 
3160 '* E$ = EXTENSION 
3170 '* 
3180 '***** 
3190 ' 
3200 DIM F$(72), E$(72) 
3210 ' 
3220 'STEP THROUGH THE 
3230 'RAW BUFFERS 
3240 FOR T1=0 TO 71 
3250 ' 
3260 'FILENAME 
3270 F$(T1)=MID$(R$(T1), 1, 8) 
3280 E$(T1)=MID$(R$(T1), 9, 3) 
3290 ' 
3300 'PROGRESS COUNTING 
3310 PRINT "*"; 
3320 IF T1=26 THEN PRINT@ 258, "*"; 
3330 IF T1=53 THEN PRINT@ 290, "*"; 
3340 NEXT T1 
3350 PRINT " -- DONE" 
3360 ' 
3500 '***** 
3510 '* 
3520 '* SELECT FILE TO READ 
3530 '* 
3540 '* G$ = 8.3 FILENAME 
3550 '* HL = FILENAME LENGTH 
3560 '* IL = EXT LENGTH 
3570 '* H$ = PADDED FILENAME 
3580 '* I$ = PADDED EXT 
3590 '* T2,T3 = TEMP VARS 
3600 '* 

3610 '***** 
3620 ' 
3630 PRINT@ 66, "          " 
3640 PRINT@ 130, "             " 
3650 PRINT@ 386, "(PRESS ANY KEY TO CONTINUE)"  
3660 A$=INKEY$ 
3670 IF A$="" GOTO 3660 
3680 ' 
3690 CLS 
3700 PRINT@ 66, "PLEASE ENTER FILENAME.EXT" 
3710 PRINT@ 130, "  " 
3720 INPUT G$ 
3730 L=LEN(G$) 
3740 'CHECK FOR FILENAME 
3750 'INPUT ERRORS 
3760 IF L>12 GOTO 4500 
3770 IF L<1 GOTO 4500  
3780 FOR I=1 TO L 
3790 IF MID$(G$,I,1)="." GOTO 3840 
3800 NEXTI 
3810 GOTO 3900 
3820 'GOTDOT 
3830 'FILENAME LENGTH 
3840 HL=I-1 
3850 'EXT LENGTH 
3860 JL=L-HL-1 
3870 GOTO 3950 
3880 'NODOT 
3890 'FILENAME LENGTH 
3900 HL=L 
3910 'EXT LENGTH 
3920 JL=0 
3930 'NAME LENGTHS  
3940 'VALIDITY CHECK 
3950 IF HL<1 GOTO 4500 
3960 IF HL>8 GOTO 4500 
3970 IF JL<0 GOTO 4500 
3980 IF JL>3 GOTO 4500 
3990 'SET INPUT FILENAME 
4000 'AND EXT 
4010 H$=LEFT$(G$,HL) 
4020 J$=RIGHT$(G$,JL) 
4030 'FILENAME AND  
4040 'EXT PADDING 
4050 IF HL>=8 GOTO 4100 
4060 T2=8-HL 
4070 FOR T3=0 TO T2-1 

4080 H$=H$+" " 
4090 NEXT T3 
4100 IF JL>=3 GOTO 4150 
4110 T2=3-JL 
4120 FOR T3=0 TO T2-1 
4130 J$=J$+" " 
4140 NEXT T3 
4150 ' 
4170 'STEP THROUGH THE 
4180 'DIRECTORY ENTRIES 
4190 FOR T1=0 TO 71 
4200 ' 
4210 'CHECK FILENAME  
4220 'FIRST CHARACTER 
4230 'VALIDITY 
4240 T2=ASC(F$(T1)) 
4250 IF T2=0 GOTO 4320 
4260 IF T2=255 GOTO 4550 
4270 ' 
4280 IF H$<>F$(T1) GOTO 4320 
4290 IF J$<>E$(T1) GOTO 4320 
4300 'FILENAME MATCH 
4310 GOTO 4350 
4320 NEXT T1 
4330 'FILE NOT FOUND 
4340 GOTO 4550 
4350 'FILEMATCH 
4360 OPEN "I",#1,G$ 
4370 IF EOF(1)=-1 GOTO 4430 
4380 CLS 
4390 PRINT@ 130, "FILE ";G$ 
4400 PRINT@ 162, "WAS SUCCESSFULLY OPENED." 
4410 PRINT@ 386, "(PRESS ANY KEY  TO CONTINUE)" 
4420 GOTO 4760 
4430 CLOSE #1 
4440 'EMPTY FILE ERROR 
4450 PRINT "EMPTY FILE ERROR" 
4460 PRINT "SYSTEM ABORT" 
4470 GOTO 32767 
4480 ' 
4490 'FILENAME INPUT ERROR 
4500 PRINT "FILENAME ERROR" 
4510 PRINT "SYSTEM ABORT" 
4520 GOTO 32767 
4530 ' 
4540 'FILE NOT FOUND ERROR 
4550 PRINT "FILE NOT FOUND" 

4560 PRINT "SYSTEM ABORT" 
4570 GOTO 32767 
4580 ' 
4700 '***** 
4710 '* 
4720 '* OUTPUT CHOICE SCREEN 
4730 '* 
4740 '***** 
4750 ' 
4760 CLS 
4770 PRINT@ 66, "PLEASE CHOOSE:" 
4780 PRINT@ 130, "(1) LINE-BY-LINE" 
4790 PRINT@ 162, "     OUTPUT TO SCREEN" 
4800 PRINT@ 226, "(2) CONTINUOUS" 
4810 PRINT@ 258, "     OUTPUT TO PRINTER" 
4820 A$=INKEY$ 
4830 IF A$="" GOTO 4820 
4840 IF A$="1" GOTO 5100 
4850 IF A$="2" GOTO 6070 
4860 PRINT@ 388, "(PLEASE PRESS 1 OR 2)" 
4870 GOTO 4820 
4880 ' 
5000 '***** 
5010 '* 
5020 '* LINE-BY-LINE 
5030 '* OUTPUT TO SCREEN 
5040 '* 
5050 '* X$ = LINE OF TEXT  
5060 '*       INPUT FROM FILE 
5070 '* 
5080 '***** 
5090 ' 
5100 CLS 
5110 PRINT 
5120 PRINT "  PRESS ANY KEY TO BEGIN, AND" 
5130 PRINT "  ANY KEY AFTER EACH LINE OF" 
5140 PRINT "  OUTPUT TO THE SCREEN." 
5150 PRINT 
5160 A$=INKEY$ 
5170 IF A$="" GOTO 5160 
5180 PRINT G$ 
5190 PRINT 
5200 LINE INPUT #1, X$ 
5210 PRINT X$ 
5220 A$=INKEY$ 
5230 IF A$="" GOTO 5220 
5240 'GO IF THERE IS MORE DATA IN FILE 

5250 IF EOF(1)=0 GOTO 5200 
5260 PRINT 
5270 PRINT "END OF FILE" 
5280 GOTO 9160 
5290 ' 
6000 '***** 
6010 '* 
6020 '* CONTINUOUS 
6030 '* OUTPUT TO PRINTER 
6040 '* 
6050 '***** 
6060 ' 
6070 CLS 
6080 PRINT 
6090 PRINT "  READY THE PRINTER, AND" 
6100 PRINT "  PRESS ANY KEY TO BEGIN" 
6110 PRINT "  PRINTING." 
6120 PRINT 
6130 A$=INKEY$ 
6140 IF A$="" GOTO 6130 
6150 PRINT#-2, G$ 
6160 PRINT#-2, " " 
6170 LINE INPUT #1, X$ 
6180 PRINT#-2, X$ 
6190 'GO IF THERE IS MORE DATA IN FILE 
6200 IF EOF(1)=0 GOTO 6170 
6210 PRINT#-2, " " 
6220 PRINT#-2, "END OF FILE" 
6230 PRINT 
6240 PRINT "END OF FILE" 
6250 GOTO 9160 
6260 ' 
9120 '* CLOSE AND EXIT 
9130 '* 
9140 '***** 
9150 ' 
9160 'CLOSE THE FILE 
9170 CLOSE #1 
9180 'RESET THE DRIVE 
9190 DRIVE 0 
9900 '***** 
9910 '* 
32767 END 
 

===== 

Appendix A: New BDS Software License  
 

This New Software License applies to all software found on the BDS Software site, and 
supersedes all previous copyright notices and licensing provisions which may appear in the 
software itself or in any documentation therefor. 

 
All software which has previously been placed in the public domain remains in the public 

domain. 
 
All other software, programs, experiments and reports, documentation, and any other 

material on this site (other than that attributed to outside sources) is hereby copyright © 2018 (or 
later if so marked) by M. David Johnson. 

 
All software, documentation, and other information on the BDS Software site is available 

for you to freely download without cost. 
 
Whether you downloaded such items directly from this site, or you obtained them by any 

other means, you are hereby licensed to copy them, to sell or give away such copies, to use them, 
and to excerpt from them, in any way whatsoever, so long as nothing you do with them would 
denigrate the name of our Lord and Savior, Jesus Christ. 

 
I make absolutely no warranty whatsoever for any of these items. You use them entirely 

at your own risk. 
 
If they don't work for you, I commiserate. 
 
If they crash your system, I sympathize. 
 
But I accept no responsibility whatsoever for any such consequences. Under no 

circumstances will BDS Software or M. David Johnson be liable for any negative results of any 
kind which you may experience from downloading or using these items. 

 
BDS Software's former mail address at P.O. Box 485 in Glenview, IL is no longer valid. 

Any mail sent to that address will be rejected by the U.S. Postal Service. See my Contact page. 
 
M.D.J. 2018/06/08 
 
===== 

 
 


